(******************************************************************************

    Пример 4

******************************************************************************)

//- DB2 контейне для a, b растояния c, отсортированный по c --------------------

DATA_BLOCK DB2

STRUCT

            val_a : ARRAY[1..10] OF INT;           // значение  a

            val_b : ARRAY[1..10] OF INT;           // значение  b

            val_c : ARRAY[1..10] OF REAL;          // растояние c

END_STRUCT 

BEGIN

END_DATA_BLOCK

//- DB1 контейнер для случ. значений a и b -------------------------------------

DATA_BLOCK DB1

  STRUCT

    buffer_a : ARRAY[1..10] OF INT := 10(0);       // несортированное значение a

    buffer_b : ARRAY[1..10] OF INT := 10(0);       // несортированное значение b

  END_STRUCT

BEGIN

END_DATA_BLOCK

//- FC62 сохранение длины ------------------------------------------------------

FUNCTION FC62 : VOID

CONST

    BUFLENGTH := 10;                     // длина буфера

END_CONST

VAR_IN_OUT

    US_VALUES:                           // IN: несортированные значения

        STRUCT                           // структура содержит массивы для          

          val_a : ARRAY[1..BUFLENGTH] OF INT;   // значений a   

          val_b : ARRAY[1..BUFLENGTH] OF INT;   // значений b

          val_c : ARRAY[1..BUFLENGTH] OF REAL;  // длины c

        END_STRUCT;

END_VAR

VAR_TEMP

    i,tmp_a,tmp_b : INT;

    tmp_c         : REAL;

    swapped       : BOOL;

END_VAR

BEGIN

    REPEAT                  // отсортированные a, b и c, в зависимости от c

        swapped := FALSE;

        FOR i := BUFLENGTH TO 2 BY -1 DO

          IF US_VALUES.val_c[i-1] > US_VALUES.val_c[i]

           THEN                       // замена позиций a, b и c в буфере

            tmp_a               := US_VALUES.val_a[i];

            tmp_b               := US_VALUES.val_b[i];

            tmp_c               := US_VALUES.val_c[i];

            US_VALUES.val_a[i]  := US_VALUES.val_a[i-1];

            US_VALUES.val_b[i]  := US_VALUES.val_b[i-1];

            US_VALUES.val_c[i]  := US_VALUES.val_c[i-1];

            US_VALUES.val_a[i-1]:= tmp_a;

            US_VALUES.val_b[i-1]:= tmp_b;

            US_VALUES.val_c[i-1]:= tmp_c;

            swapped             := TRUE;     

          END_IF;

        END_FOR;

    UNTIL NOT swapped    // повторять пока все значения не отсртированы

    END_REPEAT;

DB2 := US_VALUES;     // Имена массивов в пределах DB должны совпадать

                                                     // c именами полей из US_VALUES 
END_FUNCTION         // Иначе должны были бы копироваться массивы по отдельности 

//- FC61 вычисление длины -------------------------------------------------

FUNCTION FC61 : VOID

AUTHOR  : HF

NAME    : Abstand

FAMILY  : Training

// KNOW_HOW_PROTECT

VAR_TEMP

    values: STRUCT                   // временная структура для значений 

               val_a : ARRAY[1..10] OF INT;

               val_b : ARRAY[1..10] OF INT;

               val_c : ARRAY[1..10] OF REAL;

            END_STRUCT;

    index : INT;   

END_VAR

BEGIN

    values.val_a := DB1.buffer_a;    // взять а из DB

    values.val_b := DB1.buffer_b;    // взять b из DB

    index := 1;                      // инициализация индекса

    WHILE index <= 10 DO             // вычисление длины

        values.val_c[index]:= SQRT(SQR(values.val_a[index]) +

                              SQR(values.val_b[index]));

        index:=index +1;

    END_WHILE;

    FC62(US_VALUES := values);       // FC2 сортирует a, b и c по c

END_FUNCTION

//- FB2 сохраняет значения в буфере (DB1 ) -------------------------------------

FUNCTION_BLOCK FB2                   // случ. знач. a помещается в DB1.buffer_a 

                                     // случ. знач. b помещается в DB1.buffer_b 

CONST

    BUFLENGTH  := 10;                // длина буфера

END_CONST

VAR_INPUT

    NewValue  : INT;                 // случ. значение

END_VAR

VAR_IN_OUT

    BUF_DB    : ANY;                 // номер DB и область значений 

END_VAR

VAR_OUTPUT

    READY   : BOOL:=FALSE;           // все значения сохранены

END_VAR

VAR

    index  : INT := 1;               // cчетчик № сохраненных значений

    field  : ARRAY[1..10] OF INT;

    ret    : INT;                    // RET_VAL для SFC20

END_VAR

BEGIN

  IF index > BUFLENGTH THEN          // очиска буфера перед стартом 

        index := 1;

        READY := FALSE;

    END_IF;

    field[index]:= NewValue;         // сохранение значений

    IF index = BUFLENGTH THEN        // буфер полон

        ret   := SFC20(SRCBLK:=field,DSTBLK:=BUF_DB); // сохранить значения в DB

        READY := TRUE;               // все значения сохранены

    END_IF;

    index := index +1;

END_FUNCTION_BLOCK

//- FC60 генератор случ. чисел -------------------------------------------------

FUNCTION FC60 : INT; 

CONST

    m:=101;                          // случ. число от 0 to 100

END_CONST

VAR_INPUT

    RAND:INT;                        // инициализация

END_VAR

BEGIN                                // вычисление случ. числа

    FC60 := (ABS(RAND*DINT_TO_INT(TIME_TO_DINT(SFC64())))) MOD m;

END_FUNCTION

//- OB 35 ---------------------------------------------------------------------

ORGANIZATION_BLOCK OB35                                

VAR_TEMP

    info  : ARRAY[0..19] OF BYTE;

    random: INT;

END_VAR

BEGIN

    random := FC60(RAND:=711);                        //создание случ. числа a

    FB2.DB11(NewValue:=random,BUF_DB:=DB1.buffer_a);  //сохранение случ. числа

    random := FC60(RAND:=13);                         // создание случ. числа b

    FB2.DB12(NewValue:=random,BUF_DB:=DB1.buffer_b);  // сохранение случ. числа

    IF DB12.READY = TRUE THEN                         //если буфер полон,

        FC61();                                       //вычислить длину
    END_IF;

END_ORGANIZATION_BLOCK
