(******************************************************************************

    Пример 5

******************************************************************************)

//- DB3 контейнер для площадей и среднего значения -----------------------------

DATA_BLOCK DB3       

STRUCT

            area : ARRAY[1..10] OF REAL;            // области 

END_STRUCT 

BEGIN

END_DATA_BLOCK

//- DB2 контейнер значений a, b и длины c, сортированные по c ------------------

DATA_BLOCK DB2

STRUCT

            val_a : ARRAY[1..10] OF INT;           // значение a

            val_b : ARRAY[1..10] OF INT;           // значение b

            val_c : ARRAY[1..10] OF REAL;          // длина c

END_STRUCT 

BEGIN

END_DATA_BLOCK

//- DB1 контейнер случ. значений a и b ----------------------------------------

DATA_BLOCK DB1

  STRUCT

    buffer_a : ARRAY[1..10] OF INT := 10(0);        // несортированные a

    buffer_b : ARRAY[1..10] OF INT := 10(0);        // несортированные b

  END_STRUCT

BEGIN

END_DATA_BLOCK

//- FC63 вычисление площади ----------------------------------------------------

FUNCTION FC63 : VOID

AUTHOR  : HF

NAME    : Flaeche

FAMILY  : Training

// KNOW_HOW_PROTECT

CONST

    BUFLENGTH := 10;           // длина буфера

END_CONST

VAR_TEMP

    i        : INT;                          // индекс цикла

    val_a    : ARRAY[1..BUFLENGTH] OF INT;   // временный массив для a  

    val_b    : ARRAY[1..BUFLENGTH] OF INT;   // временный массив для b 

    val_area : ARRAY[1..BUFLENGTH] OF REAL;  // временный массив  

END_VAR

BEGIN

    val_a := DB2.val_a;

    val_b := DB2.val_b;

    FOR i := 1 TO BUFLENGTH BY 1 DO                // вычисление всех площадей

        val_area[i] := val_a[i] * val_b[i] /2;

    END_FOR;

    DB3.area := val_area;

END_FUNCTION

//- FC62 сохранение длин ------------------------------------------------------

FUNCTION FC62 : VOID

CONST

    BUFLENGTH := 10;                     // длина буфера

END_CONST

VAR_IN_OUT

    US_VALUES:                           // IN: несортированные значения

        STRUCT                           // структура массивов для          

          val_a : ARRAY[1..BUFLENGTH] OF INT;   // значений a   

          val_b : ARRAY[1..BUFLENGTH] OF INT;   // значений b

          val_c : ARRAY[1..BUFLENGTH] OF REAL;  // длин c

        END_STRUCT;

END_VAR

VAR_TEMP

    i,tmp_a,tmp_b : INT;

    tmp_c         : REAL;

    swapped       : BOOL;

END_VAR

BEGIN

    REPEAT                  // отсортированные a, b и c, по c

        swapped := FALSE;

        FOR i := BUFLENGTH TO 2 BY -1 DO

          IF US_VALUES.val_c[i-1] > US_VALUES.val_c[i]

           THEN    // swap position of a, b and c in buffer

            tmp_a               := US_VALUES.val_a[i];

            tmp_b               := US_VALUES.val_b[i];

            tmp_c               := US_VALUES.val_c[i];

            US_VALUES.val_a[i]  := US_VALUES.val_a[i-1];

            US_VALUES.val_b[i]  := US_VALUES.val_b[i-1];

            US_VALUES.val_c[i]  := US_VALUES.val_c[i-1];

            US_VALUES.val_a[i-1]:= tmp_a;

            US_VALUES.val_b[i-1]:= tmp_b;

            US_VALUES.val_c[i-1]:= tmp_c;

            swapped             := TRUE;     

          END_IF;

        END_FOR;

    UNTIL NOT swapped    // повторять, пока все значения не отсортированы

    END_REPEAT;

DB2 := US_VALUES;    // Имена массивов в пределах DB должны совпадать 

                     // c именами полей из US_VALUES 
END_FUNCTION         // Иначе должны были бы копироваться массивы по отдельности 

//- FC61 вычисление длин -------------------------------------------------

FUNCTION FC61 : VOID

AUTHOR  : HF

NAME    : Abstand

FAMILY  : Training

// KNOW_HOW_PROTECT

VAR_TEMP

    values: STRUCT                   // временная структура для значений

               val_a : ARRAY[1..10] OF INT;

               val_b : ARRAY[1..10] OF INT;

               val_c : ARRAY[1..10] OF REAL;

            END_STRUCT;

    index : INT;   

END_VAR

BEGIN

    values.val_a := DB1.buffer_a;    // взять а из DB

    values.val_b := DB1.buffer_b;    // взять b из DB

    index := 1;                      // инициализация индекса

    WHILE index <= 10 DO             // вычисление расстояния

        values.val_c[index]:= SQRT(SQR(values.val_a[index]) +

                              SQR(values.val_b[index]));

        index:=index +1;

    END_WHILE;

    FC62(US_VALUES := values);       // FC2 сортирует a, b и c по c

END_FUNCTION

//- FB2 сохранение значений в буфере (DB1 ) -----------------------------------
FUNCTION_BLOCK FB2                   // запись a в DB1.buffer_a 

                                     // запись b в DB1.buffer_b 

CONST

    BUFLENGTH  := 10;                // длина буфера

END_CONST

VAR_INPUT

    NewValue  : INT;                 // случ. знач.

END_VAR

VAR_IN_OUT

    BUF_DB    : ANY;                 //номер DB и область для значений

END_VAR

VAR_OUTPUT

    READY   : BOOL:=FALSE;           // все значения сохранены

END_VAR

VAR

    index  : INT := 1;               // счетчик № сохраненых значений

    field  : ARRAY[1..10] OF INT;

    ret    : INT;                    // RET_VAL для SFC20

END_VAR

BEGIN

  IF index > BUFLENGTH THEN          // очистка буфера на старте 

        index := 1;

        READY := FALSE;

    END_IF;

    field[index]:= NewValue;         //сохранение значений

    IF index = BUFLENGTH THEN        // буфер полон

        ret   := SFC20(SRCBLK:=field,DSTBLK:=BUF_DB); //сохранение значений в DB

        READY := TRUE;               // все значения сохранены

    END_IF;

    index := index +1;

END_FUNCTION_BLOCK

//- FC60 генератор случайных чисел ---------------------------------------------
FUNCTION FC60 : INT; 

CONST

    m:=101;                          // случайное число от 0 to 100

END_CONST

VAR_INPUT

    RAND:INT;                        // инициализация

END_VAR

BEGIN                                // вычисление случ. числа

    FC60 := (ABS(RAND*DINT_TO_INT(TIME_TO_DINT(SFC64())))) MOD m;

END_FUNCTION

//- OB 35 ---------------------------------------------------------------------

ORGANIZATION_BLOCK OB35                                

VAR_TEMP

    info  : ARRAY[0..19] OF BYTE;

    random: INT;

END_VAR

BEGIN

    random := FC60(RAND:=711);                        //создние случ.числа a

    FB2.DB11(NewValue:=random,BUF_DB:=DB1.buffer_a);  //сохранение значения

    random := FC60(RAND:=13);                         //создние случ.числа b

    FB2.DB12(NewValue:=random,BUF_DB:=DB1.buffer_b);  // сохранение значения 

    IF DB12.READY = TRUE THEN                         //если буфер полон

        FC61();                                       //вычислить расстояния

    END_IF;

    IF I0.0 <> M0.1 AND I0.0 = TRUE THEN     //если положит. Фронт на I0.0

        FC63();                              //вычислить площадь
    END_IF;

    M0.1 := I0.0;


     //Старое значение I0.0 для след. цикла                                           

END_ORGANIZATION_BLOCK

